

# Лабораторна робота 8

## Синтез блоку обчислення спецфункції

**1. Мета:** оволодіти знаннями і практичними навичками по проектуванню обчислювальних блоків послідовної дії. Також даються навички програмування і відлагодження блоків послідовної дії і автоматів на мові VHDL.

### Теоретичні відомості

Блоки послідовального действия, наприклад, множення, ділення, застосовуються в найпростіших процесорах. Вони мають в кілька разів менші апаратні витрати, ніж блоки паралельної дії.

### 2. Завдання для лабораторної роботи:

- розробити функціональну схему блока, що виконує задані функції;
- протестувати блок.

**Результати виконання** оформлюються у вигляді звіту (протоколу). Звіт повинен вміщувати:

- опис заданого варіанта блока,
- хід проектування і функціональну схему блока,
- графіки сигналів, знятих при іспитах блока,
- висновки.

У всіх варіантах завдань необхідно обчислити деяку функцію  $Y=F(X)$ , яка апроксимується заданою формулою. Блок повинен мати один суматор, один блок множення і достатню кількість регістрів для зберігання проміжних результатів. Вхідні і вихідні дані представлені 16-розрядними двійковими кодами з доповненням до 2-х з комою, що встановлена після знакового розряду, тобто  $1 > X > -1$ . Розрядність внутрішніх суматора і регістрів повинна бути достатньою, щоб не було переповнення. Обчислення повинні починатись по зовнішньому сигналу START, а їхнє закінчення повинне бути вказане вихідним сигналом RDY, який стробує результат.

Варіант завдання вибирається за номером студента в списку групи з наступної таблиці 1.

Таблиця 1. Варіанти завдань

№ завд	Функція Y
1	$\sqrt{(X^2+Z^2)} = 0.9605 X  + 0.3978 Z $ при $ X  >  Z $ ; $= 0.9859 X  + 0.2327 Z $ при $ X  > 2 Z $ .
2	$\ln(1+X) = 0.00049 + 0.98248X - 0.39728X^2 + 0.10784X^3$
3	$\sin(\pi X/2) = 1.57063X - 0.64323X^3 + 0.07271X^5$
4	$\sin(\pi X/4) = 0.78539X - 0.08071X^3 + 0.00243X^5$
5	$\sin(\pi X/4)/X = 0.78540 - 0.08073X^2 + 0.00244X^4$
6	$\sin(X) = 0.99999X - 0.30825X^3 + 0.01537X^5$
7	$\sqrt{(1+X)} = 1 + X/2 - X^2/8 + X^3/16 - 5X^4/128$
8	$1/\sqrt{(1+X)} = 1 - X/2 + 3X^2/8 - 5X^3/16 + 35X^4/128$
9	$\cos(X) = (1 - 0.40528X^2)(1 - 0.04503X^2)(1 - 0.01621X^2)$
10	$\ln(x) = X - 1 - (X - 1)^2 + (X - 1)^3 - (X - 1)^4 + (X - 1)^5 - (X - 1)^6$
11	$0.5/X = Z + Z(1 - 2XZ^2)$ , де $Z = 1.2094 - (X - 1.9193)^2/4$
12	$\sqrt{(X^2 - Z^2)} = 1.1319 X  - 0.7264 Z $ при $ X  >  Z $ ; $= 1.0186 X  - 0.2729 Z $ при $ X  > 2 Z $ .
13	$\ln(1+X) = 0.99744X - 0.4712X^2 + 0.22567X^3 - 0.05875X^4$
14	$\cos(\pi X/2) = 0.9994 - 1.2228X^2 + 0.22399X^4$
15	$\cos(\pi X/4) = 0.99999 - 0.30825X^2 + 0.01537X^4$
16	$\text{tg}(X) = X + 0.33336X^3 + 0.13285X^5 + 0.05716X^7$
17	$\cos(X) = 0.99999 - 0.49925X^2 + 0.03963X^4$
18	$\sqrt{(1-X)} = 1 - X/2 - X^2/8 - X^3/16 - 5X^4/128$
19	$1/\sqrt{(1-X)} = 1 + X/2 + 3X^2/8 + 5X^3/16 + 35X^4/128$
20	$\sin(X) = X(1 - 0.10132X^2)(1 - 0.02533X^2)(1 - 0.01126X^2)$
21	$0.5/X = 2Z(15/16 + 3XZ^2(XZ^2 - 5/6))$ , де $Z = 0.8939 - 0.405X$

### 3. Приклад виконання роботи

Розглянемо приклад проектування блоку, що виконує функцію:  $\sin(X) = X - X^3/6 + X^5/120 = X - a \cdot X^3 + b \cdot X^5$ . Граф алгоритму показано на рис.1. Алгоритм має 5 операцій множення і 2 операції додавання, не приймаючи до уваги додавання з 0. Для обчислень знадобляться суматор з регістром S, блок множення з регістром P, регістри для зберігання квадрата X2, куба X3 і результату Y. Стан вказаних регістрів при виконанні алгоритма показано в табл.2. Вхідне дане X зразу завантажується в регістр суматора S, де зберігається протягом 5 тактів і бере участь в операції додавання в 5-му такті.

Таблиця 2. Стан регістрів блоку при виконанні алгоритму.

№ такту	Суматор S	Блок множення P	Регістр X2	Регістр X3	Регістр Y
1	X				
2	X	$X^2 = X * X$			
3	X	$X^3 = X^2 * X$	X <sup>2</sup>		
4	X	$X^5 = X^2 * X^3$		X <sup>3</sup>	
5	X	$b * X^5$		X <sup>3</sup>	
6	$X + b * X^5$	$a * X^3$			
7	$X + b * X^5 - a * X^3$				
8					Y

Об'єкт блоку виглядає як наступний

```
library IEEE;
use IEEE.STD_LOGIC_1164.all;
use IEEE.STD_LOGIC_arith.all;
entity X_Y_SIN is port(CLK : in STD_LOGIC;    --синхросерія
RST : in STD_LOGIC; --сигнал початкового встановлення
START : in STD_LOGIC;    --сигнал запуску обчислень
X : in STD_LOGIC_VECTOR(15 downto 0); --вхідне дане
Y : out STD_LOGIC_VECTOR(15 downto 0); --вихідне дане
RDY : out STD_LOGIC;    --сигнал готовності результату
end X_Y_SIN;
```

Архітектура блоку описується як наступна

```
architecture X_Y_SIN of X_Y_SIN is
constant a:SIGNED(15 downto 0):= --константа a
SIGNED(CONV_STD_LOGIC_VECTOR(integer(1.0/6.0*2.0**15),16));
constant b:SIGNED(15 downto 0):= --константа b
SIGNED(CONV_STD_LOGIC_VECTOR(integer(1.0/120.0*2.0**15),16));
signal s:SIGNED(16 downto 0);--adder
signal p:SIGNED(31 downto 0);--multiplier
signal x2,x3:SIGNED(15 downto 0);--intermediate results
signal ct2:natural range 0 to 7;    --лічильник станів
begin
FSM:process(CLK,RST) begin
if RST='1' then
ct2<=6; RDY<='0';
elsif CLK='1' and CLK'event then
if ct2=7 then
RDY<='1'; -- calculation is ready
end if;
if START='1' then
ct2<=0;    --start of calculations
```




Рис.1. Граф алгоритму

```

        RDY<='0';
    elsif ct2<7 then
        ct2<=ct2+1;
    end if;
end if;
end process;
RALU:process(CLK,RST) begin
    if RST='1' then s<=(others=>'0');
        p<=(others=>'0'); x2<=(others=>'0');
        x3<=(others=>'0'); Y<=(others=>'0');
    elsif CLK='1' and CLK'event then
        case ct2 is
            when 0=> s<= signed(SXT((X),17));--input datum X
            when 1=> p<= s(15 downto 0)*s(15 downto 0);--X^2
            when 2=> p<= p(30 downto 15)*s(15 downto 0);--X^3
                x2<=p(30 downto 15);          -- X^2
            when 3=> p<= p(30 downto 15)*x2;      --X^5
                x3<=p(30 downto 15);          -- X^3
            when 4=> p<= p(30 downto 15)*b;      --b*X^5
            when 5=> p<= x3*a;                  --a*X^3
                s<= s + p(31 downto 15);          --X+b*X^5
            when 6=> s<= s - p(31 downto 15);    -- X+b*X^5-a*X^3
            when others=> Y<= STD_LOGIC_VECTOR(s(15 downto 0)); --result Y
        end case;
    end if;
end process;
end X_Y_SIN;

```

Тут константи  $a$  і  $b$  задаються як 16-бітні вектори, які є цілими числами зі знаком і які представляють дробні числа  $1/6$  і  $1/120$ , відповідно, вони мають масштабний коефіцієнт  $2^{-15}$ . Константи одержуються з реальних чисел при множенні їх на коефіцієнт  $2^{15}$  з послідовним перекладом в ціле число, 16-розрядний вектор і вектор типу SIGNED.

Результат множення  $P$  вибрано 32-розрядним, тобто рівним сумі розрядності операнда і константи-множника. У наступних розрахунках з добутку  $P$  вибираються 16 старших розряди, крім найстаршого, так як добуток чисел зі знаком має 2 однакових знакових розряди. Сума  $S$  є 17-розрядною, тобто вона має 1 додатковий розряд щоб запобігти переповненню. Всі внутрішні сигнали і константи вибрані типу SIGNED, щоб компілятор розпізнавав їх як значення зі знаком в доповнюючому коді.

Процес FSM описує керуючий автомат. Його робота полягає в запуску лічильника  $CT2$  по сигналу  $START$ , відліку 7 тактів і зупинці  $CT2$  в стані 7, в якому автомат видає сигнал готовності результату  $RDY$ .

Процес RALU виконує власне обчислення згідно з алгоритмом в табл.2. При цьому оператор **case** визначає дії в кожному такті, які відмічає лічильник  $CT2$ . Так, в нульовому такті накопичувальний суматор  $S$  записує вхідне дане  $X$ , в першому такті суматор  $S$  видає  $X$  на вхід блоку множення (див. табл.2) і блок множення одержує добуток  $P=X*X$ , який видається ним весь другий такт. В другому такті цей добуток переписується в регістр  $X2$ , а блок множення одержує добуток  $P=(X*X)*X$  і т.д. В останньому такті при  $CT2=7$  результат, що накопичився в  $S$ , переписується в регістр результату  $Y$ , який підключено до вихідного порту блоку, де зберігається, поки не буде розраховано результат для наступного вхідного даного. Структура одержаного блоку показана на рис.2.

Поведінков частина опису стенду для іспитів представлена нижче. Процес GEN генерує початкові дані  $X$  і  $START$ , а процес ERR обчислює помилку між справжнім значенням  $Y_{ref}=\sin(X)$  і обчисленим значенням  $Y$ . Графіки сигналів обчислення одного результату показані на рис. 3, а с результатів для серії даних – на рис.4.

```

begin
RST<='1', '0' after 33 ns;
CLK<=not CLK after 5 ns;
GEN: process(clk) begin
  if RISING_EDGE(CLK) then
    ctt<=ctt+1;
    if ctt mod 10 = 0 then
      X<=signed(X)+200;
      START<='1';
    else
      START<='0';
    end if;
  end if;
end process;

```

```

UUT : x_y_sin port map
(CLK => CLK,
RST => RST,
START => START,
X => X,
Y => Y,
RDY => RDY );

```

```

Yref<=CONV_STD_LOGIC_VECTOR(integer(2.0**15*SIN(
  real(CONV_INTEGER(SIGNED(X)))/2.0**15)),16);

```

```

ERR:process(RDY) begin
  if RISING_EDGE(RDY) then
    error<=signed(Y) - signed(Yref);
  end if;
end process;

```

end TB\_ARCHITECTURE;




Рис.2. Структура блока обчислення  $Y=F(X)$




Рис.3. Сигнали моделі при обчисленні одного результату




Рис.4. Вихідний результат і його похибка, що не переважає 7 молодших розрядів